Overview of Data Management Tasks

(command file=data_mgmt_lecture.sas)
Make a copy of a SAS data set:

You can use a set statement to make a copy of a data set. In the commands below, NEWMARCH is created by making an exact copy of MARCH (which we assume was created as a temporary data set in the current SAS session). 
data newmarch;

   set march;
run;

Additional commands can be added to the data step to create new variables, or to modify the data set in other ways. 

data newmarch;

   set march;

   /*additional SAS statements*/

run;

NB: be sure all changes that you wish to make to your new data set are included before the run statement! After the run statement, the data set will be closed, and no additional variables can be added, or changes made to the data set. 
A set statement can also be used to make a copy of a permanent SAS data set:

libname b510 "c:\documents and settings\kwelch\desktop\b510";

data b510.employee2;

     set b510.employee;

run;
Create a subset of data:
You can easily create a subset of your data by using the set statement along with a subsetting if statement. The subsetting if statement acts as a gateway for allowing observations to be written to a data set. In the examples below, the data set named MARCH15 will contain information on flights only on March 15th, 1990, while the data set named LONDON will contain information on all flights to London, and the data set named LONGFLT will contain information on all flights of 1000 miles or more..

data march15;

   set march;

   if date = “15MAR1990”D;

run;

data london;

   set march;

   if dest=”LON”;

run;

data longflt;

   set march;

   if miles >=1000;

run;

NB: The subsetting if can be used at any place in your data step code. It will only take effect when the data set is written out. 

Another way to select cases to be included in a data set is to use an output statement. It is important to note that the output statement takes effect immediately (at the point in your code where it is included). Any commands that are added after the output statement will not affect the cases that were output earlier.

WRONG:

data london_latemarch;

   set march;

   if dest=”LON” and date >=”15MAR1990”D then output;

   totpassngrs = boarded + transfer + nonrev;
   pctfull = (totpassngrs/capacity)*100;
run;

RIGHT:

data london_latemarch2;

   set march;

   totpassngrs = boarded + transfer + nonrev;
   pctfull = (totpassngrs/capacity)*100;

   if dest=”LON” and date >=”15MAR1990”D then output;

run;

The data set LONDON_LATEMARCH will contain flights to London on or after March 15th, but the new variables TOTPASSNGRS and PCTFULL will be included in the new data set, but they will not have any values in them, because they were defined after the output statement.
Delete CASES from a data set:
A delete statement can be used to remove a case or cases from a data set. When the case is deleted, it is permanently removed from the data set. The delete statement takes effect immediately when it is specified, so deleted cases will not be available for any later programming statements.

data shortflt;

   set march;

   if miles >=1000 then delete;
   if date=. then delete;

run;
Similar to the output statement, the delete statement takes effect at the point in the data step where it is placed.
Keep or Drop VARIABLES:
You can control the variables that are included in a SAS data set by using keep and drop statements as part of the data step. The keep and drop statements may be given at any point in the data step, and only take effect at the time the data set is written. 

data march_passngrs;

   set march;

   keep date time orig dest miles boarded transfer nonrev deplane capacity;run;

data march_passngrs2;

   set march;

   drop mail freight;

run;

Create new variables using transformations and recodes:

New variables can be created using transformations and recodes of variables in the data step using assignment statements with SAS functions, or if…then statements. Assignment statements (shown below) are used to create new variables based on the value of previously defined variables, expressions, or constants, and are of the form: 

newvar = expression;

The example below shows how to create new variables using a data step:

data march_recode;

   set march;

   totpassngrs = boarded + transfer + nonrev;

   totpassngrs2 = sum(boarded,transfer,nonrev);

   logpassngrs = log(totpassngrs);

   empty_seats = capacity - totpassngrs;

   totnonpass  = sum(mail,freight);

   pctfull = (totpassngrs/capacity)*100;

   int_pctfull = int(totpassngrs/capacity)*100;

   rnd_pctfull = round(pctfull,.1); 

   if pctfull = 100 then full_flight = 1;

   else full_flight = 0;

   if pctfull = . then full_flight = .;

   if pctfull not=. then do;

      if pctfull < 25 then full_cat = 1;

      if pctfull >=25 and pctfull <50 then full_cat=2;

      if pctfull >=50 and pctfull <75 then full_cat=3;

      if pctfull >=75 then full_cat=4;

   end;

   if dest = "CPH" or dest="FRA" or dest = "LON" 

     or dest = "PAR" or dest = "YYZ" then USA = 0;

   if dest in("DFW", "LAX", "ORD", "WAS") then USA = 1;

   obama_date = "20JAN2009"D;

   format obama_date mmddyy10.;

   elapsed_time = obama_date - date;

run;  
Check the new variables using Proc Means and Proc Freq:

title "Check New Variables";

proc means data=march_recode;

run;

proc freq data=march_recode;

  tables full_flight full_cat dest USA;

run;

                                      Check New Variables   07:14 Tuesday, January 20, 2009   5

                                      The MEANS Procedure

Variable        Label         N            Mean         Std Dev         Minimum         Maximum

-----------------------------------------------------------------------------------------------

DATE            DATE        634        11031.98       8.9801263        11017.00        11047.00

MILES           MILES       635         1615.25         1338.47     229.0000000         3857.00

MAIL            MAIL        634     381.0031546      74.6288128     195.0000000     622.0000000

FREIGHT         FREIGHT     634     333.9511041      98.1122248      21.0000000     631.0000000

BOARDED         BOARDED     633     132.3570300      43.4883098      13.0000000     241.0000000

TRANSFER        TRANSFER    635      14.4062992       5.3362008               0      29.0000000

NONREV          NONREV      635       4.1133858       1.9243731               0       9.0000000

DEPLANE         DEPLANE     635     146.7842520      45.4289656      18.0000000     250.0000000

CAPACITY        CAPACITY    635     205.3795276      27.1585929     178.0000000     250.0000000

totpassngrs                 633     150.8878357      43.0930520      31.0000000     250.0000000

totpassngrs2                635     150.4598425      43.6959260       9.0000000     250.0000000

logpassngrs                 633       4.9681880       0.3292127       3.4339872       5.5214609

empty_seats                 633      54.5244866      34.9192529               0     151.0000000

totnonpass                  635     713.8283465     127.0958820     218.0000000         1085.00

pctfull                     633      73.0774908      17.7696598      17.2222222     100.0000000

int_pctfull                 633       8.3728278      27.7198922               0     100.0000000

rnd_pctfull                 633      73.0764613      17.7693610      17.2000000     100.0000000

full_flight                 633       0.0837283       0.2771989               0       1.0000000

full_cat                    633       3.3791469       0.6785651       1.0000000       4.0000000

USA                         632       0.6819620       0.4660832               0       1.0000000

obama_date                  635        17917.00               0        17917.00        17917.00

elapsed_time                634         6885.02       8.9801263         6870.00         6900.00

----------------------------------------------------------------------------------------------
                                      Check New Variables      

                                       The FREQ Procedure

                                                        Cumulative    Cumulative

                full_flight    Frequency     Percent     Frequency      Percent

                ----------------------------------------------------------------

                          0         580       91.63           580        91.63

                          1          53        8.37           633       100.00

                                     Frequency Missing = 2

                                                      Cumulative    Cumulative

                 full_cat    Frequency     Percent     Frequency      Percent

                 -------------------------------------------------------------

                        1           4        0.63             4         0.63

                        2          59        9.32            63         9.95

                        3         263       41.55           326        51.50

                        4         307       48.50           633       100.00

                                     Frequency Missing = 2

                                   Destination City

                                                     Cumulative    Cumulative

                    dest    Frequency     Percent     Frequency      Percent

                    ---------------------------------------------------------

                    CPH           27        4.26            27         4.26

                    DFW           62        9.78            89        14.04

                    FRA           27        4.26           116        18.30

                    LAX          123       19.40           239        37.70

                    LON           58        9.15           297        46.85

                    ORD           92       14.51           389        61.36

                    PAR           27        4.26           416        65.62

                    PRD            1        0.16           417        65.77

                    QAS            1        0.16           418        65.93

                    WAS          154       24.29           572        90.22

                    YYZ           62        9.78           634       100.00

                                      Frequency Missing = 1

                                                   Cumulative    Cumulative

                    USA    Frequency     Percent     Frequency      Percent

                    --------------------------------------------------------

                      0         201       31.80           201        31.80

                      1         431       68.20           632       100.00

                                      Frequency Missing = 3

The two variables, TOTPASSNGRS and TOTPASSNGRS2 have different numbers of cases, because they were created in different ways. TOTPASSNGRS is created using the mathematical operators (+), so the resulting variable is missing if any of the variables in the expression is missing. TOTPASSNGRS2 returns the sum of the non-missing argument variables, so if any of the argument variables has a value, the result will have a value.

Also note the syntax used to create the new variables, FULL_FLIGHT and FULL_CAT. When using “Else” with SAS, all other values, including missing will be included in the Else category. We get around this by setting cases with a missing value for PCTFULL to missing in the resulting variable. We use an If…then statement when creating FULL_CAT to be sure this new variable is only created if PCTFULL is not missing (if pctfull not=. then do;). If you use an if…statement, it must be followed by an end statement. 

The new variable USA is created from the character variable, DEST. The in operator is used to shorten the syntax for setting up the value of USA=1. 
Working with dates:

The example below shows how you can read in dates using different methods, and work with dates to create AGE. Dates in SAS are automatically stored as numeric variables, containing the number of days since Jan 1, 1960.

data dates;

input dategrad mmddyy10. pulse1 pulse2 bmonth bday byear;

 avgpulse = mean(pulse1,pulse2);

 new_avgpulse = (pulse1+pulse2)/2;

 bdate = mdy(bmonth,bday,byear);

 current_date = "20JAN2009"D;

  format bdate mmddyy10.;

  format current_date mmddyy10.;

  format dategrad mmddyy10.;

 age1 = (current_date - bdate)/365.25;

 age2 = int(age1);

 agegrad = (dategrad-bdate)/365.25;

 cards;

06/12/1993 72 85 6 15 1975

05/30/2000 68 .  10 3 1982

 ;

proc print data=dates;

run;

 1  06/12/1993  72  85   6  15  1975  78.5  78.5  06/15/1975  01/20/2009  33.6016  33  17.9932

Sort Cases:
A data set is sorted using Proc Sort. Once sorted, a data set remains sorted, and any later analyses can be done either for the entire data set, or for subgroups by including a by statement in a given procedure.

proc sort data=march_recode;

  by USA;

run;

title “Descriptive Statistics by US vs Non-US Destinations”;

proc means data=march_recode;

  by USA;

run;

A separate analysis will be done for each of the "by" groups, even missing. To avoid this, use a where statement.

proc means data=march_recode;

  by USA;

  where USA not=.; 

run;

                     Descriptive Statistics by US vs Non-US Destinations                     

-------------------------------------------- USA=. --------------------------------------------

                                      The MEANS Procedure

Variable        Label         N            Mean         Std Dev         Minimum         Maximum

-----------------------------------------------------------------------------------------------

DATE            DATE          3        11029.67       5.5075705        11024.00        11035.00

MILES           MILES         3         1148.00         1177.28     229.0000000         2475.00

MAIL            MAIL          3     363.3333333     101.7857226     247.0000000     436.0000000

FREIGHT         FREIGHT       3     301.3333333     134.8196326     187.0000000     450.0000000

BOARDED         BOARDED       3     140.6666667      33.0050501     104.0000000     168.0000000

TRANSFER        TRANSFER      3      16.3333333       6.6583281      12.0000000      24.0000000

NONREV          NONREV        3       2.6666667       2.3094011               0       4.0000000

DEPLANE         DEPLANE       3     153.6666667      26.6333125     123.0000000     171.0000000

CAPACITY        CAPACITY      3     200.0000000      17.3205081     180.0000000     210.0000000

totpassngrs                   3     159.6666667      24.8260616     132.0000000     180.0000000

totpassngrs2                  3     159.6666667      24.8260616     132.0000000     180.0000000

logpassngrs                   3       5.0645842       0.1618285       4.8828019       5.1929569

empty_seats                   3      40.3333333      39.0683162               0      78.0000000

totnonpass                    3     664.6666667     214.0895451     434.0000000     857.0000000

pctfull                       3      80.7936508      18.6039601      62.8571429     100.0000000

int_pctfull                   3      33.3333333      57.7350269               0     100.0000000

rnd_pctfull                   3      80.8000000      18.5841330      62.9000000     100.0000000

full_flight                   3       0.3333333       0.5773503               0       1.0000000

full_cat                      3       3.6666667       0.5773503       3.0000000       4.0000000

obama_date                    3        17917.00               0        17917.00        17917.00

elapsed_time                  3         6887.33       5.5075705         6882.00         6893.00

-----------------------------------------------------------------------------------------------

-------------------------------------------- USA=0 --------------------------------------------

Variable        Label         N            Mean         Std Dev         Minimum         Maximum

-----------------------------------------------------------------------------------------------

DATE            DATE        200        11031.95       9.0577142        11017.00        11047.00

MILES           MILES       201         2630.47         1524.04     366.0000000         3857.00

MAIL            MAIL        200     385.3850000      74.0259820     213.0000000     606.0000000

FREIGHT         FREIGHT     200     332.9450000      97.1224269      65.0000000     598.0000000

BOARDED         BOARDED     201     153.1940299      37.4436532      71.0000000     241.0000000

TRANSFER        TRANSFER    201      13.7910448       5.5078235               0      29.0000000

NONREV          NONREV      201       4.3084577       1.8985200               0       9.0000000

DEPLANE         DEPLANE     201     171.2935323      37.2003818      88.0000000     250.0000000

CAPACITY        CAPACITY    201     227.7910448      33.3367083     178.0000000     250.0000000

totpassngrs                 201     171.2935323      37.2003818      88.0000000     250.0000000

totpassngrs2                201     171.2935323      37.2003818      88.0000000     250.0000000

logpassngrs                 201       5.1181690       0.2301583       4.4773368       5.5214609

empty_seats                 201      56.4975124      34.4502720               0     147.0000000

totnonpass                  201     714.7562189     126.4874906     218.0000000         1061.00

pctfull                     201      75.6035105      14.3174987      41.2000000     100.0000000

int_pctfull                 201       5.9701493      23.7524547               0     100.0000000

rnd_pctfull                 201      75.6034826      14.3175500      41.2000000     100.0000000

full_flight                 201       0.0597015       0.2375245               0       1.0000000

full_cat                    201       3.4726368       0.5661250       2.0000000       4.0000000

obama_date                  201        17917.00               0        17917.00        17917.00

elapsed_time                200         6885.06       9.0577142         6870.00         6900.00

-----------------------------------------------------------------------------------------------

-------------------------------------------- USA=1 --------------------------------------------

Variable        Label         N            Mean         Std Dev         Minimum         Maximum

-----------------------------------------------------------------------------------------------

DATE            DATE        431        11032.02       8.9757925        11017.00        11047.00

MILES           MILES       431         1145.05     921.8176570     229.0000000         2475.00

MAIL            MAIL        431     379.0928074      74.8317227     195.0000000     622.0000000

FREIGHT         FREIGHT     431     334.6450116      98.5498184      21.0000000     631.0000000

BOARDED         BOARDED     429     122.5361305      42.7943542      13.0000000     207.0000000

TRANSFER        TRANSFER    431      14.6798144       5.2335711               0      29.0000000

NONREV          NONREV      431       4.0324826       1.9286971               0       9.0000000

DEPLANE         DEPLANE     431     135.3062645      44.5281613      18.0000000     210.0000000

CAPACITY        CAPACITY    431     194.9651972      15.0173913     180.0000000     210.0000000

totpassngrs                 429     141.2657343      42.4698823      31.0000000     210.0000000

totpassngrs2                431     140.6798144      43.2351551       9.0000000     210.0000000

logpassngrs                 429       4.8972431       0.3456721       3.4339872       5.3471075

empty_seats                 429      53.6993007      35.1411111               0     151.0000000

totnonpass                  431     713.7378190     127.0626962     341.0000000         1085.00

pctfull                     429      71.8400118      19.0763566      17.2222222     100.0000000

int_pctfull                 429       9.3240093      29.1108084               0     100.0000000

rnd_pctfull                 429      71.8384615      19.0759007      17.2000000     100.0000000

full_flight                 429       0.0932401       0.2911081               0       1.0000000

full_cat                    429       3.3333333       0.7223620       1.0000000       4.0000000

obama_date                  431        17917.00               0        17917.00        17917.00

elapsed_time                431         6884.98       8.9757925         6870.00         6900.00

----------------------------------------------------------------------------------------------  

Sorting by more than one variable:

You can sort by several variables, as shown in the example below. Proc sort organizes the data so that the first variable represents the slowest changing index (i.e., cases will be sorted first by DATE, and then by levels of  DEST within DATE). 

proc sort data=march_recode;

     by date USA;

run;

title “Descriptive Statistics by Date and Destination”;

proc means data=march_recode;

     by date USA;

     where USA not=.;

run;

Using the Tagsort Option:

Sorting is one of the more computationally intensive operations that can be done on a SAS data set.  It requires a lot of hard drive space, which can be a problem, especially if the data set has many observations and a large number of variables. You can be more efficient in sorting if you use the tagsort option.  This method basically sorts only the key variables and then rebuilds the dataset by pulling up the appropriate observation and attaching all the rest of the variables. This method of sorting will usually take longer, but uses less hard drive space.  

proc sort data=march_recode tagsort;

     by date dest;

run;

Creating a New Sorted Data Set:

If you wish to create a new data set, and maintain the input data set in its original order, you can use the out= option on the Proc Sort statement, as shown below:

proc sort data=march_recode out=sortdat;

     by totpassngrs;

run;

Combine data sets by adding new Cases using SET:  

data march15;

  set march;

  if date ="15Mar1990"D;

run;

data march16;

  set march;

  if date = "16Mar1990"d;

run;

data march15_16;

  set march15 march16;

run;

Combine data sets by adding new Variables using MERGE:  
A merge statement can be used to combine two (or more) data sets to add new variables. Both data sets must be sorted before merging.
data employee_demog;

  set b510.employee;

  keep id gender bdate educ jobcat minority;

run;

data employee_salary;

  set b510.employee;

  keep id salary salbegin jobtime prevexp;

run;

proc sort data=employee_demog;

  by id;

run;

proc sort data=employee_salary;

  by id;

run;

data employee_combine;

  merge employee_demog employee_salary;

  by id;

run;
How to combine data sets if you don't have the same cases:
Use the special temporary variable IN= to specify the cases that come from each data set.
data employee_longtime;

  set employee_salary;

  if  prevexp>=100;

run;

proc sort data=employee_longtime;

  by id;

run;

title "Merge All Cases";

data employee_longtime_combine;

  merge employee_demog employee_longtime;

  by id;

run;

title "Merge only Cases in both data sets";

data employee_longtime_combine;

  merge employee_demog(in=ina)

        employee_longtime(in=inb);

  by id;

  if ina and inb;

run;
Bringing in data from a table:
You may have a data set that has values that need to be matched to all cases in a certain group. SAS automatically attaches values to all cases that match when you bring in data from a table. This is a very useful feature in SAS. 
data avgsal;

  input minority avg_USsalary;

cards;

0 36000

1 24000

;

title "Average Salary from a table";

proc print data=avgsal;

run;

proc sort data=avgsal;

by minority;

run;

proc sort data=b510.employee;

by minority;

run;

data employee_avgsal;

  merge b510.employee

        avgsal;

  by minority;

run;

proc means data=employee_avgsal;

class minority;

run;

Selecting cases for Procs:

Cases can be selected for most Procs by using a where statement. You can select cases based on the value of a character or numeric variable.

Selecting cases based on values of a character variable:

title "Flights to Los Angeles";

proc print data=march_recode;

    where dest = "LAX";

    var flight dest totpassngrs;

run;

If you wish to select the observations to be included in an analysis based on a missing value for a character variable, use quotes around a blank " ", because blank is missing for a character variable.

title "Missing Destination";

proc print data=march_recode;

    where dest = " ";

    var flight dest totpassngrs;

run;

Selecting cases based on values of a numeric variable:

Cases used in an analysis may be selected based on the values of a numeric variable. The Boolean operators (<, >, <=, >=, =, ~=) may be used to get the desired case selection, as shown below:

title "Flights Less than 30 Percent Full";

proc print data=march_recode;

    where pctfull < 30;

    var dest date pctfull;

run;

Those with PCTFULL missing are also included in this case selection, because missing is evaluated as less than any numeric value.   

                              Flights Less than 30 Percent Full                             

                              Obs    dest          date    pctfull

                               99    ORD     03/05/1990    28.0952

                              102    WAS     03/05/1990    17.2222

                              235    DFW     03/12/1990    27.2222

                              390    WAS     03/19/1990    29.4444

                              421    LAX     03/21/1990      .

                              451    WAS     03/22/1990    18.3333

                              512    WAS     03/25/1990    23.8889

                              539    DFW     03/27/1990      .

                              635    WAS     03/31/1990    18.8889

To make sure you do not include the missing values in your case selection, you can use syntax like that shown below:

title "Flights Less than 30 Percent Full";
title2 "Exclude Missing";

proc print data=march_recode;

    where pctfull not=. and pctfull < 30;

    var dest date pctfull;

run;

If you wish to select observations based on a missing value for a numeric variable, use a period to indicate missing, as shown in the example below.  

title "Cases Where Number of Passengers is Missing";

proc print data=march_recode;

    where totpassngrs = .;

    var flight dest totpassngrs;

run;

The where statement can also be used with “between” to restrict cases used in an analysis. The example below will print those cases with percent full from 25 to 35:

title "Flights Between 25 and 35 Percent Full";

proc print data=march_recode;

    where pctfull between 25 and 35;
run;

You can also select cases using a combination of character and numeric variables in the where statement:

title "Flights less than 60 percent full to London";

proc print data=march_recode;

    where (pctfull < 60) and (dest="LON") ;

    var flight dest totpassngrs capacity pctfull;
run;

Selecting cases based on dates:

You can select cases for a procedure based on dates, by using a SAS date constant.  Note that the date constant is specified in quotes with the day as a two-digit number, followed by a three-letter abbreviation for the month, followed by a 2 or 4-digit number for the year. A letter D (either upper or lower case) must appear after the quote to let SAS know that this is a date, and therefore numeric, and not a character value.

title "Flights on March 7th, 1990"; 

proc print data=march_recode;

    where date = "07MAR90"D;

run;

You can also use "where … between" with dates to specify a range of dates:

title "Flights March 7th to March 9th , 1990"; 

proc print data=march_recode;

    where date between "07MAR90"D and "14MAR90"D;

run;

You can use the same method for selecting observations based on missing values for a date variable as for a numeric variable, because dates are stored simply as numeric values in SAS.

title "Cases with Missing Date";

proc print data=march_recode;

    where date = .;


    var flight dest date;

run;

